

Az vevő- és vezérlőáramkör programja

; Központizár-vezérlés - IR vevő- és vezérlőáramkör

INCLUDE 89C2051.mc

```
*****  
;  
;  
;  
;  
;  
;  
;  
;  
*****
```

VÁLTOZÓK

BYTECNT	EQU	70h	; Bájt számláló
RXB	EQU	71h	; Vett bájt
STATUS	EQU	72h	; Soros vétel státusza
SZINKRON	EQU	73h	; Szinkronkarakter állapota
ERROR	EQU	74h	; Hibás-vett-bájt számláló
CNT1	EQU	75h	; Késleltető számláló
CNT2	EQU	76h	; Késleltető számláló
CNT3	EQU	77h	; Késleltető számláló
DATA	EQU	78h	; ADC adat
CNT4	EQU	79h	; ADC időzítő
STATE	EQU	69h	; Rendszerállapot

```
*****  
;  
;  
;  
;  
;  
;  
;  
;  
*****
```

PORTOK

RAMP	BIT	P1.0	; Fűrészjel-bemenet (komparátorbemenet)
VIN	BIT	P1.1	; Analóg bemenet (komparátorbemenet)
IEN	BIT	P1.2	; Fűrészjel-engedélyezés kimenet
DRV_B	BIT	P1.3	; A teljesítményhíd "B" meghajtójele
DRV_A	BIT	P1.4	; A teljesítményhíd "A" meghajtójele
D2	BIT	P1.5	; D0...D2 vezérlőjel-kimenetek
D1	BIT	P1.6	
D0	BIT	P1.7	
RXD	BIT	P3.0	; Soros bemenet
SWL	BIT	P3.1	; "Bal oldali zár nyitva" jel
SWR	BIT	P3.2	; "Jobb oldali zár nyitva" jel
CLOSE	BIT	P3.3	; "Ajtók becsukva" jel
IRR	BIT	P3.4	; IR-vevő engedélyezés (negált)
LED	BIT	P3.5	; LED-kimenet
COMPAR	BIT	P3.6	; A komparátor kimenete
BEEPER	BIT	P3.7	; Beeper kimenet

```
*****  
;  
;  
;  
;  
;  
;  
;  
;  
*****
```

VEKTORTÁBLA

ORG 000h ; Reset

LJMP BEGIN

ORG 23h ; Serial Interrupt
LJMP RECBYTE

```
.....  
;  
;  
;  
;  
;  
;  
.....  
;
```

FŐPROGRAM

ORG 0100h

BEGIN:

LCALL INIT ; Rendszer-inicializálás

LOOP1: ; Inaktív állapot (nyitva)
JNB SWR,LOOP1_1 ; Ha a jobb oldal záródik, akkor zárás
JNB SWL,LOOP1_1 ; Ha a bal oldal záródik, akkor zárás
LJMP LOOP1_2

LOOP1_1:
LCALL ZARAS

LOOP1_2:
MOV A,STATE ; Az aktuális állapot beolvasása
CJNE A,#1,LOOP1 ; Ha STATE != 1, akkor nyitva van
LJMP LOOP2

LOOP2: ; Aktív állapot (zárva)

LCALL Wait ; 1,2 s várakozás

LCALL Wait

LCALL Wait

LCALL Wait

LCALL Wait

LCALL Wait

CLR LED ; LED bekapcsolása

MOV CNT1,#127

LCALL Wait1 ; 0,6 s várakozás

SETB LED ; LED kikapcsolása

LCALL ADC ; ADC elindítása - telepfeszültség mérése

LCALL UCHECK ; Telepfeszültség ellenőrzése

MOV A,STATE ; Az aktuális állapot beolvasása

CJNE A,#0,LOOP2 ; Ha STATE != 0, akkor zárva van

LJMP LOOP1

```
.....  
;  
;  
;  
;  
;  
;  
.....  
;
```

RENDSZERINICIALIZÁLÁS

INIT:

CLR IRR ; Az infravevő bekapcsolása

CLR IEN ; Fűrészjel letiltása

CLR DRV_A ; DRV_A kimenet kikapcsolása

CLR DRV_B ; DRV_B kimenet kikapcsolása

MOV SP,#07h ; Stack Pointer a kezdőállapotba

MOV IE,#0h ; Összes interrupt-bit letiltása

MOV TMOD,#20h ; Timer 1 [80C51 HW manual 85. old.]

MOV TH1,#0F2h ; Reload érték [80C51 HW manual 97. old.]

ANL PCON,#00h	; SMOD1 = 0, SMOD0 = 0
MOV SCON,#40h	; Uart Mode 1, REN = 0 [80C51 HW manual 92. old.]
SETB TR1	; Timer 1 Run
MOV BYTECNT,#10	; Bájt számláló alapállapot
MOV ERROR,#10	; Hibaszámláló alapállapot
CLR LED	; LED bekapcsolása
LCALL BEEP	; Hangjelzés
LCALL Wait	; Várakozás
SETB LED	; LED kikapcsolása
SETB ES	; Serial interrupt engedélyezése
SETB EA	; Global interrupt engedélyezése
SETB REN	; Soros vétel engedélyezése
RET	

```

*****
;
;
;
;
;
;
;
*****

```

SOROS VONALI INTERRUPT

```

RECBYTE:           ; A soros bemeneten érkező bájtok kiértékelése
  CLR REN           ; Soros vevő letiltása
  JB RI,REC1        ; Ha RI = 0, akkor nem Rec. Int.
  LJMP RECEND
REC1:
  MOV RXB,SBUF      ; Vett adat az RXB regiszterbe
  MOV A,#0          ; Szinkron karakter: 00h
  CJNE A,RXB,REC2   ; Ha RXB!=00h, akkor REC2
  MOV SZINKRON,#1   ; Ha RXB=00h, akkor szinkron karakter érkezett
  LJMP RECEND
REC2:
  MOV A,SZINKRON    ; Nem szinkron karakter érkezett
  CJNE A,#0h,REC3   ; Ha már volt szinkron karakter, akkor REC3
  LJMP RECEND
REC3:
  MOV A,BYTECNT     ; Ha BYTECNT! = 10, akkor REC6
  CJNE A,#10,REC6   ; Ha BYTECNT = 10
  MOV DPTR,#ZARKOD ; A zárás kódszó 10. bájtjának beolvasása A-ba
  MOVC A,@A+DPTR    ; Ha RXB! = ZARKOD(10), akkor REC4
  CJNE A,RXB,REC4

  MOV STATUS,#0h    ; STATUS = 0
  DEC BYTECNT       ; BYTECNT = BYTECNT-1
  MOV ERROR,#10     ; ERROR = 10
  LJMP RECEND
REC4:               ; BYTECNT = 10 és RXB! = ZARKOD(10);
  MOV A,BYTECNT
  MOV DPTR,#NYITKOD
  MOVC A,@A+DPTR    ; A nyitás kódszó 10. bájtjának beolvasása A-ba
  CJNE A,RXB,REC5   ; Ha RXB! = NYITKOD(10), akkor REC5

  MOV STATUS,#1h    ; STATUS = 1
  DEC BYTECNT       ; BYTECNT = BYTECNT-1
  MOV ERROR,#10     ; ERROR = 10
  LJMP RECEND
REC5:               ; BYTECNT=10 és RXB!=ZARKOD(10) és RXB! =
                   ; NYITKOD(10)
  MOV SZINKRON,#0
  DJNZ ERROR,RECEND ; Ha ERROR! = 0, akkor RECEND

```

```

LCALL Wait ; Biztonsági várakozás a próbálgatások kiszűrésére
MOV ERROR,#10
LJMP RECEND
REC6: ; BYTECNT != 10
MOV A,STATUS
CJNE A,#0h,REC8 ; Ha STATUS != 0, akkor REC8

MOV A,BYTECNT
MOV DPTR,#ZARKOD
MOVC A,@A+DPTR ; A zárás kódszó köv. bájtyának beolvasása A-ba
CJNE A,RXB,REC7 ; Ha RXB! = ZARKOD(BYTECNT), akkor REC7

DJNZ BYTECNT,RECEND ; BYTECNT = BYTECNT-1, ha BYTECNT != 0, akkor
; RECEND
LCALL ZARAS ; Ha BYTECNT = 0, akkor ZÁRÁS
MOV BYTECNT,#10
MOV SZINKRON,#0
LJMP RECEND
REC7:
MOV BYTECNT,#10
MOV SZINKRON,#0
DJNZ ERROR,RECEND ; Ha ERROR! = 0, akkor RECEND
LCALL Wait ; Biztonsági várakozás a próbálgatások kiszűrésére
MOV ERROR,#10
LJMP RECEND
REC8: ; BYTECNT != 10 és STATUS != 0
MOV A,BYTECNT
MOV DPTR,#NYITKOD
MOVC A,@A+DPTR ; A nyitás kódszó köv. bájtyának beolvasása A-ba
CJNE A,RXB,REC7 ; Ha RXB! = NYITKOD(BYTECNT), akkor REC7
DJNZ BYTECNT,RECEND ; BYTECNT = BYTECNT-1, ha BYTECNT != 0, akkor
; RECEND
LCALL NYITAS ; Ha BYTECNT = 0, akkor NYITÁS
MOV BYTECNT,#10
MOV SZINKRON,#0
LJMP RECEND
RECEND:
CLR RI
SETB REN
RETI

```

```

*****
;
;
;
;
;
;
;
*****

```

ZÁRVEZÉRLÉS

```

ZARAS:
JNB CLOSE,ZARAS1 ; Ha az ajtó nyitva van (CLOSE = 0), akkor hibajelzés
CLR LED ; LED bekapcsolása
SETB DRV_A ; A teljesítményhíd vezérlése
LCALL BEEP ; Hangjelzés
LCALL Wait ; Várakozás
LCALL BEEP ; Hangjelzés
CLR DRV_A ; A teljesítményhíd kikapcsolása
LCALL Wait ; Várakozás
LCALL BEEP ; Hangjelzés
MOV STATE,#1 ; Állapot = 1 (zárva)
LJMP ZARAS2

```

```

ZARAS1:                                ; Ha CLOSE = 0, akkor hiba (nem zárja be)
    LCALL BEEP5
ZARAS2:                                ; LED kikapcsolása
    SETB LED
    RET

NYITAS:
    JNB CLOSE,NYITAS1                  ; Ha az ajtó nyitva van (CLOSE = 0), akkor hibajelzés
    SETB DRV_B                          ; A teljesítményhíd vezérlése
    LCALL BEEP                          ; Hangjelzés
    LCALL Wait                          ; Várakozás
    LCALL BEEP                          ; Hangjelzés
    CLR DRV_B                          ; A teljesítményhíd kikapcsolása
    MOV STATE,#0                       ; Állapot = 0 (nyitva)
    LJMP NYITAS2
NYITAS1:                                ; Ha CLOSE = 0, akkor hiba (nem nyitja ki)
    LCALL BEEP5                          ; Hangjelzés
NYITAS2:
    RET

```

```

*****
;
;
;
;
;
;
;
;
;
;
;
*****

```

BEEPER SZUBRUTINOK

```

BEEP:                                    ; Rövid hangjelzés
    MOV CNT3,#0FFh                      ; BEEPER periódusszámláló
BEEP0:
    CLR BEEPER                          ; Négyszögjel-generálás (T = 500µs)
    LCALL SWait
    SETB BEEPER
    LCALL SWait
    DJNZ CNT3,BEEP0
    RET

BEEP1:                                    ; Hosszú hangjelzés
    LCALL BEEP
    LCALL BEEP
    LCALL BEEP
    LCALL BEEP
    RET

BEEP5:                                    ; 5 rövid hangjelzés
    LCALL BEEP
    LCALL Wait
    LCALL BEEP
    LCALL Wait
    LCALL BEEP
    LCALL Wait
    LCALL BEEP
    LCALL Wait
    LCALL BEEP
    RET

```

```

*****
;
;
;
;
;
;
;
;
;
;
;
*****

```

KÉSLELTETŐ SZUBRUTINOK

```

Wait:                ; 112206 ciklusnyi késleltetés (a meghívással együtt)
                    ; (kb. 200ms @ 6,76438 MHz)
                    ;  $T = (2+2+(2+CNT2*2+2)*CNT1+2)*Tc = \text{max. } 131076*Tc$ 
MOV CNT1,#255       ; [2]
Wait1:
MOV CNT2,#218       ; [2]
Wait2:
DJNZ CNT2,Wait2     ; [2]
DJNZ CNT1,Wait1     ; [2]
RET                  ; [2]

```

```

Swait:               ; 140 ciklusnyi késleltetés (kb. 250 μs @ 6,76438MHz)
                    ;  $T = (2+2+CNT2*2+2)*Tc = \text{max. } 516*Tc$ 
MOV CNT2,#67        ; [2]
SWait1:
DJNZ CNT2,SWait1   ; [2]
RET                  ; [2]

```

```

;*****
;
;
;
;
;
;
;*****
;

```

A/D-KONVERZIÓ

```

ADC:
MOV CNT4,#0FFh     ; Számláló (időzítő) kezdeti érték
CLR IEN             ; A fűrészel letiltása
CLR RAMP            ; A C6 kondenzátor kisütése
LCALL SWait        ; Rövid várakozás
SETB RAMP           ; A RAMP bemenet engedélyezése
SETB IEN            ; A fűrészel engedélyezése
NOP                 ; [1]
NOP                 ; [1]
ADC1:
NOP                 ; [1]
NOP                 ; [1]
JB COMPAR,ADCEND   ; [2] Ha komparált, akkor vége a konverzióknak
DJNZ CNT4,ADC1     ; [2] Ha nem, akkor számláló-dekrementálás

```

```

ADCEND:
CLR IEN             ; A fűrészel letiltása
MOV A,CNT4          ; A számláló értéke A-ba
CPL A               ; Bitenkénti invertálás
MOV DATA,A        ; Az A értéke a DATA regiszterbe
RET

```

```

;*****
;
;
;
;
;
;
;*****
;

```

TÁPFESZÜLTSG ELLENŐRZÉS ÉS POWER-DOWN VEZÉRLÉS

```

UCHECK:
MOV A,DATA          ; A tápfeszültség ellenőrzése
CLR C               ; A mért érték A-ba
CJNE A,#193,UCHECK ; C flag törlése
                    ; Ha A < 193, akkor C < -1

```

```

UCHECK0:
  JNC UCHECK1                ; Ha C = 0, akkor rendben (A >= 193)
  LCALL BEEP1                 ; Ha C = 1, akkor (A < 193)
  LCALL PWRDWN                ; PWRDWN szubrutin
UCHECK1:
  RET

```

```

PWRDWN:                        ; Power Down mód bekapcsolása
  SETB IRR                    ; Infravevő kikapcsolása
  ORL PCON,#00000010b         ; Power Down - kikapcsolás
  NOP
  NOP
  NOP
  RET

```

```

*****
;
;
;
;
;
;
;
*****

```

RENDSZERKÓDOK

```

ORG 500h
ZARKOD:
  DB " ",10h,20h,30h,40h,50h,60h,70h,80h,90h,0A0h,00h ; A zárás kódja + szink. kar.
NYITKOD:
  DB " ",0A0h,90h,80h,70h,60h,50h,40h,30h,20h,10h,00h ; A nyitás kódja + szink. kar.

```

```

ORG 600h
  DB "2005.12.20. Hodány Ákos"
END

```